Effects of Kinetic Friction on the Run−time of a Bead Sliding Along a Catenary Under Gravity Pull

Haiduke Sarafian

Pennsylvania State University
1031 Edgecomb Ave., York, PA 17403
has2@psu.edu

We consider two points of unequal abscissas on a vertical plane. We pin the ends of a soft and uniform chain of length *l* at these points. Assuming the length of the chain is longer than the distance between the points, the chain under gravity pull assumes a Catenary shape. In conjunction with this setting, we work on two projects. In the first one, with the help of *Mathematica*, we exercise two mathematical scenarios; leaving one of the pins intact, we move the second pin to a set of selected positions on the same plane without changing a) the ordinance and, b) the abscissa. Utilizing *Mathematica,* we animate both processes. In the second project, we release a bead from the mobile pin. Applying *Mathematica*, we evaluate the run−time of the bead along the Catenaries of the first project. In the latter case, moreover, we consider two practical situations: a) a frictionless chain, and, b) a rough chain. We tabulate the run−time as a function of the coefficient of kinetic friction. Applying *Mathematica,* we animate the movement of the bead along the corresponding Catenaries.

■Introduction:

It is a classic problem of interest to study the geometrical characteristics associated with the assumed shape of a massive uniform line such as a laundry rope, a jump rope, a utility power line and or a flexible chain pulled by the gravity. The common features of these examples are: the lines are uniform, they are pinned at two ends in a vertical plane and that gravity is fully in action. The strategies adapted to analyze the problem falls in one of the two categories, either Newton mechanics [1] or the Calculus of Variations [2]. The end product of these two distinct analytic methods, however, is the same −− the lines assume Catenary shapes. Simply put, a hyperbolic cosine function describes the shape of the line.

Building on this classic academic background the author comes to realize its practical geometry−related applications. To begin, hold the ends of a chain, e.g. a pocket watch chain, in each hand. Keep the hands at the same height, and then, without moving the hands up or down bring them together or pull them apart. Alternatively, without changing the distance between the hands, move one of its ends either up or down. As long as the distance between the hands is shorter than the length of the chain, as mentioned, the hanging chain assumes a Catenary shape. Although it is easy to toy with the chain, it is challenging to simulate the shapes mathematically. The author applied *Mathematica*'s [3] integrated superb symbolic, numeric and graphic features to replicate the problem. Utilizing *Mathematica*'s animation, the replicated models resemble real−time cases.

Having concurred the proposed Catenary's geometry−related issues of interest, we considered a few kinematic and dynamic physics related situations. For instance, we select one of the Catenaries, e.g. the one with an uneven heights, and release a massive bead from its high point; the bead negotiates the Catenary. Applying appropriate physics principles, we evaluate the bead's run−time, the bead's horizontal and vertical speeds at any given time, and its time dependent "attack" angle. We also look into dynamic quantities such as the time dependency of the chain's surface reaction. Having established this basis, we modify the calculation including the kinetic friction.

This article is composed of two sections. In section 1, we study the proposed geometry−related issues discussed in the first part of the introduction. In section 2 we analyze the physics of the problem symbolically, conducive to a set of comprehensive formulas. Section 2 includes also numeric tabular values of quantities of interest along with corresponding graphs. We conclude the article by animating the movement of the bead. We close the article with a few conclusive remarks.

■Section 1: Geometrical Analysis

Denoting the uniform linear weight density of the line by w_0, the tension at the bottom of the line by T_0, according to [1] in a vertical upright Cartesian coordinate system with the y−axis through the minimum of the Catenary, the equation of the Catenary is given by $y(x) = \frac{T_0}{w_0} \cosh\left(\frac{w_0}{T_0} x\right)$. Since we are interested in the geometry of the problem, we suppress the physical quantities, and define a characteristic length $c = \frac{T_0}{w_0}$. Furthermore, motivated by the practical settings, we choose a coordinate system such that the vertical axis passes through the left−end of the chain; to do so, we introduce an adjustable parameter λ. We also slide the x−axis by α. In this coordinate system the Catenary becomes, $\lambda + y(x) = c \cosh(\frac{x-\alpha}{c})$. We determine the values of α and λ according to our needs.

In practice, we would be dealing with a piece of a chain with a given length l. Applying Catenary equation, symbolically, we evaluate its length in terms of its relevant parameters. To do so, we integrate a short arc length, $d\,l$, along the curve. Applying Pythagorean theorem yields, $l = \int d\,l \equiv \int \sqrt{dx^2 + dy^2} \equiv \int_0^x \sqrt{1 + y'^2}\, dx = c\,[\sinh(\frac{x-\alpha}{c}) + \sinh(\frac{\alpha}{c})]$.

The length and the coordinate of the left−end of the chain are at our disposal; we select $l = 1.4$ units and set $p_1 = \{x_{01}, y_{01}\} = \{0, 0.8 \text{ units}\}$, respectively. The mobile right−end of the chain according to the first scenario of interest described in the introduction is to move the hands back and forth without changing its height. For instance, confining its horizontal movement to $0.2 \le x_2 \le 0.8$ units, and considering the fact that the distance between the ends of the chain is to be shorter than its length, the coordinate of the mobile end becomes $p_2 = \left\{x_2,\ y_{01} + \sqrt{l^2 - x_2^{\,2}}\right\}$. This data set is entered as:

```
In[5]:= x01 = 0.; y01 = 0.8; length = 1.4;
```

In[6]:=
```
x2 = Range[0.2, 0.8, 0.2];
y2 = Range[y01, y01 + Sqrt[length^2 - Max[x2]^2], 0.2];
```

Now the challenge is to search for a set of parameters that are subject to a three coupled conditional equations,

$$\begin{cases} \lambda + y_{01} = c\cosh\left(\frac{\alpha}{c}\right) \\ \lambda + y_2 = c\cosh\left(\frac{x_2 - \alpha}{c}\right) \\ l = c\left(\sinh\left(\frac{x_2 - \alpha}{c}\right) + \sinh\left(\frac{\alpha}{c}\right)\right) \end{cases}$$

Although, one might be tempted to solve this set of equations for a set of selected coordinate p_2 one at a time, applying *Mathematica*'s FindRoot along with Table command, enables us to solve the equations in one step. Noting the fact that these equations are non−algebraic and transcendental, we apply numeric method. However, the numeric method calls for seed search values. By inspection, and by trial and error we guesstimate the range of the seed values.

In[8]:=
```
rootsx = Table[
  FindRoot[{y01 + λ == c Cosh[α/c], Min[y2] + λ == c Cosh[(x2[[n]] - α)/c],
    length == c (Sinh[α/c] + Sinh[(x2[[n]] - α)/c])}, {α, x2[[n]]/2},
   {c, 0.1}, {λ, 0.4}, MaxIterations → 200], {n, 1, Length[x2]}]
(*Table[{c Cosh[α/c]-λ,c Cosh[(x2[[n]]-α)/c]-λ,
   c (Sinh[α/c]+Sinh[(x2[[n]]-α)/c])}/.
  rootsx[[n]],{n,1,Length[x2]}]*)
```

Out[8]=
```
{{α → 0.1, c → 0.0247882, λ → -0.0995612},
 {α → 0.2, c → 0.0651546, λ → -0.0969743},
 {α → 0.3, c → 0.122945, λ → -0.0892852},
 {α → 0.4, c → 0.206932, λ → -0.0700542}}
```

Supplying the evaluated parameters we animate and display the associated Catenaries.

In[9]:=
```
Table[
  Plot[Evaluate[(c Cosh[(x - α)/c] - λ) /. rootsx[[n]]], {x, 0, x2[[n]]},
   PlotRange → {{0, 0.8}, {0, 1.4}}, AxesLabel → {"x", "y"},
   PlotLabel -> "    Catenary \n A hanging chain",
   PlotStyle → {Thickness[0.008]}], {n, 1, Length[x2]}];
```


Catenary
A hanging chain
y
1.2
1
0.8
0.6
0.4
0.2
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x


Catenary
A hanging chain
y
1.2
1
0.8
0.6
0.4
0.2
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x


Catenary
A hanging chain
y
1.2
1
0.8
0.6
0.4
0.2
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x



For sake of comparison we display collectively all the cases together.

In[10]:=
```
tabplotx = Table[Plot[Evaluate[(c Cosh[(x - α)/c] - λ) /. rootsx[[n]]],
    {x, 0, x2[[n]]}, PlotRange → {{0, 0.8}, {0, 1.4}},
    PlotLabel -> "    Catenary \n A hanging chain",
    PlotStyle → {Thickness[0.008]}, AxesLabel → {"x", "y"},
    DisplayFunction → Identity], {n, 1, Length[x2]}];
Show[tabplotx, DisplayFunction → $DisplayFunction];
```

From In[10]:=



Moving the free, right–end of the chain to the left and right results in a family of Catenaries.

Similarly, we solve the same set of the equations and search for a set of parameters corresponding to up and down movement of the right, free–end of the chain.

```
In[12]:= rootsy = Table[
    FindRoot[{y01 + λ == c Cosh[α/c], y2[[n]] + λ == c Cosh[(Max[x2] - α)/c],
       length == c (Sinh[α/c] + Sinh[(Max[x2] - α)/c])}, {α, Max[x2]/2},
      {c, 0.1}, {λ, 0.4}, MaxIterations → 200], {n, 1, Length[y2]}]
   (*{c Cosh[α/c]-λ,c Cosh[(Max[x2]-α)/c]-λ,
     c (Sinh[α/c]+Sinh[(Max[x2]-α)/c])}/.rootsy*)

Out[12]= {{α → 0.4, c → 0.206932, λ → -0.0700542},
   {α → 0.369928, c → 0.209061, λ → -0.168825},
   {α → 0.336483, c → 0.216124, λ → -0.264567},
   {α → 0.29424, c → 0.230845, λ → -0.354829},
   {α → 0.229758, c → 0.262056, λ → -0.430605},
   {α → 0.081498, c → 0.355519, λ → -0.435099}}
```

Supplying these parameters we animate and display the associated Catenaries.

```
In[13]:= Table[Plot[(c Cosh[(x - α)/c] - λ) /. rootsy[[n]], {x, 0, Max[x2]},
     PlotRange → {{0, Max[x2]}, {0, y01 + √(length^2 - Max[x2]^2)}},
     AxesLabel → {"x", "y"},
     PlotLabel -> "    Catenary \n A hanging chain",
     PlotStyle → {Thickness[0.008]}], {n, 1, Length[rootsy]}];
```




Catenary
A hanging chain
y
1.75
1.5
1.25
1
0.75
0.5
0.25
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x


Catenary
A hanging chain
y
1.75
1.5
1.25
1
0.75
0.5
0.25
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x


Catenary
A hanging chain
y
1.75
1.5
1.25
1
0.75
0.5
0.25
0.1 0.2 0.3 0.4 0.5 0.6 0.7 0.8
x





For sake of comparison we display collectively all the cases together.

In[14]:=
```
tabploty =
  Table[Plot[(c Cosh[(x - α)/c] - λ) /. rootsy[[n]], {x, 0, Max[x2]},
    PlotRange → {{0, Max[x2]}, {0, y01 + Sqrt[length^2 - Max[x2]^2]}},
    PlotLabel -> "    Catenary \n A hanging chain",
    PlotStyle → {Thickness[0.008]}, AxesLabel → {"x", "y"},
    DisplayFunction → Identity], {n, 1, Length[rootsy]}];
Show[tabploty, DisplayFunction → $DisplayFunction];
```

From In[14]:=



Moving the free, right–end of the chain up and down results in a family of Catenaries.

■**Section 2: Physical Analysis**

It is the objective of this section to analyze the impact of the characteristic curvatures of a Catenary to the movement of a massive bead traversing the Catenary's length under the gravity pull. For sake of simplicity we assume the chain while the bead is sliding along the Catenary preserves the Catenary's shape. Interested readers may relax this assumption and study its implications. Figure 3 shows the situation at hand.

One method to analyze the problem is to apply Newton's law; $\vec{F}_{\text{net}} = m\,\vec{a}$. Projecting this equation along the coordinate axes gives,



A display of a Catenary (solid curve) along with the active forces $m\,g$, N and f_k on a bead of mass m.

$$-N\sin\theta + f_k\cos\theta = m\,\ddot{x} \tag{1}$$

$$N\cos\theta + f_k\sin\theta - m\,g = m\,\ddot{y} \tag{2}$$

The relevant active forces: weight of the bead $m\,g$, the reaction of the chain N, and kinetic friction f_k are shown in Figure 3. Components of the acceleration of the bead, a_x and a_y are denoted by $\ddot{x}$ and $\ddot{y}$ respectively. The auxiliary angle θ the "attack angle" is also shown. We set $f_k = \mu\,N$ and manipulate eqs (1,2), we arrive at,

$$\frac{1+\mu\tan(\theta)}{\mu+\tan(\theta)} = \frac{g+\ddot{y}}{\ddot{x}} \tag{3}$$

We utilized angle θ to project the various components of the relevant forces along the coordinate axes. Angle θ is also the same angle that at the point of interest a tangent to the Catenary makes with respect to x−axis; we set $\tan(\theta) = y'$. With this linker concept, the geometric characteristics of the Catenary comes into the play. Equation (3) becomes,

$$\frac{1+\mu\,y'}{\mu+y'} = \frac{g+\ddot{y}}{\ddot{x}} \tag{4}$$

Applying the linker concept, we generalize our calculation and for time being put aside the specificities of the Catenaries; we replace y with $f(x)$. Noting that $f(x)$ could represent curves other than a Catenary provides opportunities to apply our methodology to analyze a vast class of similar problems. In terms of $f(x)$ we evaluate; $\dot{y}(t) \equiv \dot{x}(t)\,f'(x)$ and $\ddot{y}(t) \equiv \ddot{x}(t)\,f'(x) + \dot{x}^2(t)\,f''(x)$ with the primes denoting the spacial derivatives i.e. $f'(x) = \frac{d}{dx} f(x)$ and $f''(x) \equiv \frac{d^2}{dx^2} f(x)$. Substituting these identities in eq (4) yields,

$$\ddot{x}\left[1 + f'(x)^2\right] + \dot{x}^2\; f''(x)[f'(x) - \mu] + g\,[f'(x) - \mu] = 0 \tag{5}$$

One might realize that in the absence of kinetic friction, $\mu = 0$, eq (5) for a slanted incline with an inclination angle θ with respect to the horizontal axis reduces to the expected well known horizontal acceleration $\ddot{x} = -(g\sin\theta)\cos\theta$.

Back to our problem, eq (5) for a Catenary $f(x) \equiv y(x) + \lambda = c\cosh\left[\frac{x(t)-\alpha}{c}\right]$ yields, $f'(x) = \sinh\left[\frac{x(t)-\alpha}{c}\right]$, $f''(x) = \frac{1}{c}\cosh\left[\frac{x(t)-\alpha}{c}\right]$. Utilizing the trigonometry identity $\cosh^2\beta - \sinh^2\beta \equiv 1$, eq (5) gives,

$$\cosh^2\left[\frac{x(t)-\alpha}{c}\right]\ddot{x} + \frac{1}{c}\cosh\left[\frac{x(t)-\alpha}{c}\right]\left\{\sinh\left[\frac{x(t)-\alpha}{c}\right] - \mu\right\}\dot{x}^2 + g\left\{\sinh\left[\frac{x(t)-\alpha}{c}\right] - \mu\right\} = 0 \tag{6}$$

Equation (6) is a non−trivial, non−linear second order differential equation with variable coefficients; *Mathematica* fails to solve the equation symbolically. However, it solves it numerically with ease! Plotting and tabulating its solutions provides ample and insightful information about the kinematics and dynamics of the bead. Assuming the bead is released at rest from the high at the right−end of the chain, we supply the needed initial condition; $\dot{x}(t=0) = 0$. We wrap the Table command around the NDSolve and for a set of reasonable practical values of μ, e.g. $0 \le \mu \le 0.8$ we generate its solutions. For a demonstrative purpose, with no prejudice, we select one of the Catenaries, e.g. the one with the highest right−end. The characteristic parameters of this Catenary are given by rootsy⟦6⟧; we get,

In[16]:=
```
g = 9.80; α = 0.081498; c = 0.355519; λ = -0.435099;
```

In[17]:=
```
solαcμ =
  Table[NDSolve[{Cosh[(x[t] - α)/c]^2 x''[t] + 1/c Cosh[(x[t] - α)/c]
       (Sinh[(x[t] - α)/c] - μ) x'[t]^2 + g (Sinh[(x[t] - α)/c] - μ) == 0,
     x[0] == 0.8, x'[0] == 0}, x[t], {t, 0, 5}], {μ, 0, 0.8, 0.2}];
```

Solutions of eq (6) are $x(t)$. Knowing the time dependency of $x(t)$ helps to evaluate the vertical position of the bead in its descent as well. Applying $y(x) + \lambda = c \cosh\left[\frac{x(t)-\alpha}{c}\right]$ we plot and tabulate $\{x(t), y(t)\}$; the tables are suppressed. For the frictionless case, we have,

```
In[18]:= plotμ0 = Plot[Evaluate[x[t] /. solαcμ[[1]]], {t, 0, 0.55},
            PlotRange → {0, 2}, PlotStyle → {GrayLevel[0]},
            AxesLabel → {"t,s", "x,m"}, DisplayFunction → Identity];
         Table[{t, x[t] /. solαcμ[[1]]}, {t, 0, 0.55, 0.01}] // Short;
         y1 = c Cosh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c] - λ;
         plotμy1 = Plot[Evaluate[y1], {t, 0, 0.55}, PlotRange → {0, 2},
            PlotStyle → {Dashing[{0.02, 0.02}], GrayLevel[0]},
            AxesLabel → {"t,s", "y,m"}, DisplayFunction → Identity];
         Table[{t, c Cosh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c] - λ},
            {t, 0, 0.55, 0.01}] // Short;

In[23]:= Show[plotμ0, plotμy1, AxesLabel -> {"t,s", "x,y"},
           DisplayFunction → $DisplayFunction(*,
            GridLines→{Range[0.45,0.6,0.05], None}*)];
```

From In[23]:=



Plot of $x(t)$ (solid line) and $y(t)$ (dashed line) v.s. the run–time for a frictionless Catenary.

Figure 4 along with its tabular values (not printed) shows the bead starts freely at $x = 0.8$ and in the absence of friction slides along the entire length of the chain in 0.55 s.

Additional kinematical quantities such as v_x and v_y are also readily measured. These two are displayed in Fig 5.

```
In[24]:= Plot[Evaluate[{D[-x[t] /. solαcμ[[1]], {t, 1}], D[-y1, {t, 1}]}],
           {t, 0.001, 0.54}, AxesLabel → {"t,s", "vx ,vy "},
           PlotStyle → {{GrayLevel[0]}, {Dashing[{0.02, 0.02}]}}];
```

From In[24]:=



Plot of the horizontal (solid line) and the vertical (dashed line) speeds v.s. the run−time for a frictionless chain.

Utilizing solutions of eq (6) we plot the time dependent horizontal position of the bead for various values of μ. Similar plots, for $y(t)$ are prepared but are not included.



Plots of $x(t)$ for various values of μ. The inner (solid) and the outline (long dashed) lines correspond to $\mu = 0$ and $\mu = 0.6$; rougher the chain, longer the run−time.

Having evaluated $x(t)$ allows us to focus on a few dynamical quantities of interest, such as the chain reaction, $N(t)$. To this end, we solve eq (1) for $N = \frac{m\ddot{x}}{\cos\theta(\mu - \tan\theta)}$. Utilizing identities, $\cos\theta = \frac{1}{\sqrt{1+\tan^2\theta}}$ and $\tan\theta = y'(x)$ yields, $N = \frac{m\ddot{x}}{\mu - y'(x)}\sqrt{1 + y'(x)^2}$. Substituting for $y'(x) = \sinh\left[\frac{x(t)-\alpha}{c}\right]$ gives, $N = \frac{m\ddot{x}}{\mu - \sinh\left[\frac{x(t)-\alpha}{c}\right]}\cosh\left[\frac{x(t)-\alpha}{c}\right]$. Since in the last equation the mass of the bead is a scaling factor, without losing the generalities, we set $m = 1$. The plot of the last intext equation is

```
In[25]:= d2 = Evaluate[D[x[t] /. solαcμ[[1]], {t, 2}]];
         sh1 = Sinh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c];
         ch1 = Cosh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c];
         Plot[d2/(μ /. μ → 0. - sh1) ch1,
           {t, 0.1, 0.549}, AxesLabel → {"t,s", "N,newton"}];
```

From In[25]:=



Reaction of the chain, $N(t)$ in newtons v.s. the run−time.

As expected, at the end of the swing the reaction of the chain is at maximized.

When ever possible, it is always a good idea to cross check the accuracy of the calculation and the utilized methodology with alternative methods. To this end, we stretch our imagination and extend the Catenary at hand beyond its left−end into the negative x region. The bottom of this extended Catenary inherits the smallest curvature, a small arc segment of it can be fitted with a circular loop; one such circle along with its center coordinate {0.0814,1.2} and 0.4 unit radius is shown in fig(8). The bead at the bottom of the swing may be thought as looping in this loop. Since the weight and the surface reaction are now almost aliened, applying Newton's law gives $N = m\left(g + \frac{v^2}{R}\right)$; here v^2 is the squared speed of the bead. Knowing $\{x(t), y(t)\}$ yields $v^2 = \{v_x(t), v_y(t)\}^2$ and its value at the end of the run−time $v^2 (t \simeq 0.54) \simeq 19.77 \left(\frac{m}{s}\right)^2$. This gives $N(t = 0.54) \simeq 58$ Newtons; this is in agreement with Fig(7). *Mathematica* codes yielding to this result and the fitted loop are,

```
In[29]:= speed2 =
           Apply[Plus, Flatten[(Evaluate[{D[x[t] /. solαcμ[[1]], {t, 1}],
                 D[y1, {t, 1}]}] /. t → 0.54)^2]];
```



Display of the fitted loop and the extended Catenary.

We further our analysis by taking advantage of known $x\,(\mathrm{t})$ and now evaluate the attack angle $\theta(t)$; we apply $y'\,(x) = \tan\theta = \sinh\left[\frac{x(t)-\alpha}{c}\right]$. Solving this equation for θ yields, $\theta(t) = \arctan\left[\sinh\left(\frac{x(t)-\alpha}{c}\right)\right]$; *Mathematica* gives,

```
In[30]:= plotangleθμ0 =
   Plot[ArcTan[Sinh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c]] 180/π,
    {t, 0, 0.55}, PlotRange → {0, 80},
    AxesLabel → {"t,s", "θ,deg"}, PlotStyle → {GrayLevel[0]}];
  Table[{t, ArcTan[Sinh[(Evaluate[x[t] /. solαcμ[[1]]] - α)/c]] 180/π},
    {t, 0, 0.54, 0.01}] // Short;
```

From In[30]:=



Within the run−time of the frictionless chain, the attack angle θ plunges from 74.9° at the right−end of the chain to almost zero.

Similar graphs for non−frictionless cases are prepared but are not included.

In closing we apply *Mathematica* animation and bring some of the discussed topics together. For instance, as the bead slides along the Catenary we clock the bead and synchronize its coordinate with the associated run−time. Similar animations for non−frictionless cases are prepared but are not included.

In[32]:=
```
points = Partition[Flatten[Table[{x[t] /. solαcμ⟦1⟧,
        c Cosh[1/c (Evaluate[x[t] /. solαcμ⟦1⟧] - α)] - λ},
      {t, 0.01, 0.55, 0.02}]], 2];
listplotpoints = Table[ListPlot[{points⟦n⟧},
    PlotStyle → {GrayLevel[0.5], PointSize[0.04]},
    PlotLabel → StringJoin["t(s)=", ToString[0.01 + 0.02 n]],
    AxesLabel → {"x,m", "y,m"}, DisplayFunction → Identity],
   {n, 1, Length[points]}];
ploty6 = Show[tabploty⟦6⟧, PlotRange → {{0, 0.8}, {0, 1.8}}];
Table[Show[{listplotpoints⟦n⟧, ploty6},
   PlotRange → {{0, 0.8}, {0, 1.8}},
   DisplayFunction -> $DisplayFunction],
  {n, 1, Length[points]}];
```



■Conclusions:

We began with a classic problem, a Catenary problem. Its geometrical properties has been visited many times with variety of objectives some reviewed in [1,2]. However, there is no reference to simulations similar to the ones we presented in our work. For the first time we demonstrated two real−life practical geometrical simulations. We also proposed and analyzed a related physics dynamics problem. The challenge of the latter was heightened by including kinetic friction − a problem that its details and beauties would have remained hidden had we not applied *Mathematica*'s powerful, amazing symbolic, numeric, and graphical interrogative features. In short, we proposed a problem and we offered a research−flavored analysis. Throughout the text, on various occasions we suppressed a few investigated results. However, the *Mathematica* codes are fully embedded to assist the interested readers to produce the non−reported pieces. In the course of our investigation, on two occasions we looked into the bigger picture; we presented a generalized formulation and proposed an open− ended challenging project.

■References:

1. I.S. Sokolnikoff, and R.M. Redheffer, *Mathematics of Physics and Modern Engineering,* 2nd Ed., McGraw−Hill, Inc. New York, 1966.
2. L. Elsgolts, *Differential Equations and the Calculus of Variations*, MIR Publications, Moscow, 1970.
3. S. Wolfram, *The Mathematica Book,* 5tn Ed., Cambridge University Publication, 2003.

■About the author:

Haiduke Sarafian is John T. and Paige S. Smith Professor of Science at the Pennsylvania State University at the York campus of the Commonwealth Campuses. He received his Ph.D. in theoretical nuclear physics from Michigan State University in 1983. In 1999 he was a *Mathematica* Visiting Scholar. He is also an independent *Mathematica* Trainer, www.wolfram.com/services/training/sarafian.html

Haiduke Sarafian
Commonwealth Campuses
The Pennsylvania State University
Department of Physics
1031 Edgecomb Avenue
York, PA 17403
has2@psu.edu